ENGR 173

Name: _____________________________

Chapter 2 Part 2 Lab Activities

10 points (of 75 total)

Activity 3.  (5 pt.  Instructor’s initials ________) Walking forward – “Data table” approach
You probably noticed when typing up subroutines M0 through M14 in Activities 1 and 2 that there was a lot of repetition.  All the subroutines are essentially the same.  Compare for example subroutines M3 and M4.  

	M3:

  FOR Pulses = RightTilt TO LeftTilt STEP TiltStep

    PULSOUT TiltServo, Pulses

    PULSOUT StrideServo, LeftForward

    PAUSE MoveDelay

  NEXT

  RETURN
	M4:

  FOR Pulses = LeftForward TO RightForward STEP StrideStep

    PULSOUT TiltServo, LeftTilt

    PULSOUT StrideServo, Pulses

    PAUSE MoveDelay

  NEXT

  RETURN


The subroutines are practically identical, differing only in the beginning and end values sent to the tilt and stride servos.  In fact each of 15 subroutines written so far can be differentiated by only 4 things: (1) The beginning and (2) end points of the tilt servo rotation, and (3) the beginning and (4) end points of the stride servo rotation.  The idea of replacing 15 subroutines with a single subroutine having 4 inputs is attractive.  
We still need those 4 variables in order to describe each of the primitive movements.  One way to provide them is to store their values in EEPROM, then access them when needed.  We will use DATA and READ statements to do this.  Wang and Gracey call this the “Data table” approach.
1. Load and run the following program. 
‘************************************************************

'Program FirstStepsForwardUsingTable

' {$STAMP BS2}

' {$PBASIC 2.5}

'-------I/O Definitions ---------------

StrideServo CON       12

TiltServo   CON       13

'--------- Constant Declarations ------------

TiltStep  CON       5

StrideStep CON      5

MoveDelay CON       25

RightTilt CON       620

CenterTilt CON      750

LeftTilt CON        880

RightForward CON 650

StrideCenter CON 750

LeftForward  CON 850

'-----------Variable Declarations ----------

MoveLoop VAR Nib    'loop variable for normal walking

Pulses VAR Word

BasicMoveEEPROM VAR Word   'EEPROM location pointer for basic movement descriptions

PrimitiveMoveIndex VAR Byte    'Movement number (e.g. 1 = tilt from center to right, keep feet underneath)

PrimitiveMoveEEPROM VAR Word  'EEPROM location pointer for primitive movement descriptions

TiltStart VAR Word            'starting pulsewidth to give Tilt servo

TiltEnd VAR Word              'ending pulsewidth for TiltServo

StrideStart VAR Word

StrideEnd VAR Word

'-----------EEPROM Data ------------------

'The following basic movements are composed of the listed primitive movements

StartForward DATA 1, 2, 255        'list of primitive movements for startup

WalkForward DATA 3, 4, 5, 6, 255  'primitive moves for normal walking

FinishForward DATA 3, 7, 8, 255    'primitive movements for ending

'Data for individual primitive movements - each move described by 4 numbers:

' start of tilt, end of tilt, start of stride, end of stride

M1 DATA Word CenterTilt, Word RightTilt, Word StrideCenter, Word StrideCenter

M2 DATA Word RightTilt, Word RightTilt, Word StrideCenter, Word LeftForward

M3 DATA Word RightTilt, Word LeftTilt, Word LeftForward, Word LeftForward

M4 DATA Word LeftTilt, Word LeftTilt, Word LeftForward, Word RightForward

M5 DATA Word LeftTilt, Word RightTilt, Word RightForward, Word RightForward

M6 DATA Word RightTilt, Word RightTilt, Word RightForward, Word LeftForward

M7 DATA Word LeftTilt, Word LeftTilt, Word LeftForward, Word StrideCenter

M8 DATA Word LeftTilt, Word CenterTilt, Word StrideCenter, Word StrideCenter

'----- Main --------------

Main:

  GOSUB M0        'center both servos

  BasicMoveEEPROM = StartForward    'Start walking, startup movements to get into swing of things

  GOSUB GenericMove

  FOR MoveLoop = 1 TO 3  'take one full step each loop

    BasicMoveEEPROM = WalkForward

    GOSUB GenericMove

  NEXT

  BasicMoveEEPROM = FinishForward    'come to a stop with feet underneath

  GOSUB GenericMove

END

'------Subroutines

M0:     'initial act - center both servos

  FOR Pulses = 1 TO 100 STEP StrideStep

    PULSOUT TiltServo, CenterTilt

    PULSOUT StrideServo, StrideCenter

    PAUSE MoveDelay

  NEXT

  RETURN

GenericMove:             'Step through movements listed in BasicMoveEEPROM

  READ BasicMoveEEPROM, PrimitiveMoveIndex

    BasicMoveEEPROM = BasicMoveEEPROM + 1    'increment the pointer to the EEPROM location for the next primitive

  IF PrimitiveMoveIndex = 255 THEN RETURN  '255 marks the end of the data, therefore return to main

  LOOKUP PrimitiveMoveIndex, [M1, M1, M2, M3, M4, M5, M6, M7, M8], PrimitiveMoveEEPROM  'notice that M1 is repeated,

                                                                   'so there are 9 entries in table labeled 0 through 8

                                                                   'Index 0 is never used

   READ PrimitiveMoveEEPROM, Word TiltStart, Word TiltEnd, Word StrideStart, Word StrideEnd

  IF TiltStart = TiltEnd THEN

    FOR Pulses = StrideStart TO StrideEnd STEP StrideStep

      PULSOUT TiltServo, TiltStart

      PULSOUT StrideServo, Pulses

      PAUSE MoveDelay

    NEXT

  ELSE

    FOR Pulses = TiltStart TO TiltEnd STEP TiltStep

      PULSOUT TiltServo, Pulses

      PULSOUT StrideServo, StrideStart

      PAUSE MoveDelay

    NEXT

  ENDIF

  GOTO GenericMove         'loop until PrimitiveMoveIndex = 0
‘*****************************************************************
Things to notice:
1. We’ve defined two kinds of movements: primitive movements, and basic movements.  Primitive movements are the most fundamental.  For example, M1 is nothing more than a right tilt with the feet remaining directly under the robot.  Basic movements, in contrast, are sequences of primitive movements having a specific goal.  For example, WalkForward is a single step forward, and is composed of primitive movements M3 (tilt from right to left), M4 (move right foot forward), M5 (tilt from left to right), and M6 (move left foot forward).  In later projects we will string together basic movements to form advanced movements, like 90° turns and figure-8s.

2. The first 3 DATA instructions are:

StartForward DATA 1, 2, 255
WalkForward DATA 3, 4, 5, 6, 255
FinishForward DATA 3, 7, 8, 255

In the first DATA instruction, StartForward acts as the label for EEPROM location 0.  The number 1 is stored there.  The number 2 is stored in location 1, and 255 in location 2.  In the second DATA instruction, WalkForward is the label assigned to EEPROM location 3, and the number 3 is stored there.  And so forth.  Since there are 12 total numbers stored, 12 bytes of EEPROM (locations 0 through 11) are used.
3. The next DATA instruction is: 

M1 DATA Word CenterTilt, Word RightTilt, Word StrideCenter, Word StrideCenter 

M1 is the label for EEPROM locations 12.  The Word modifier for each value means 2 bytes will be required to store it.  For example, since CenterTilt = 750, RightTilt = 620, and StrideCenter = 750, the number 750 is stored in EEPROM locations 12 and 13, the number 620 is stored in locations 14 and 15, 750 is stored in 16 and 17, and 750 is stored again in 18 and 19.  M2, being the label for the next DATA instruction, identifies EEPROM location 20.
4. The READ instruction has the following format: READ EEPROM Location, Variable.  In program FirstStepsUsingTable, the READ instructions are located in subroutine GenericMove.  The first is: 

READ BasicMoveEEPROM, PrimitiveMoveIndex

The first time GenericMove is called, BasicMoveEEPROM has been set equal in the main routine to StartForward, which we have already seen points to EEPROM location 0.  So READ BasicMoveEEPROM, PrimitiveMoveIndex reads the number 1 from EEPROM location 0 and assigns it to variable PrimitiveMoveIndex.
5. LOOKUP is an instruction we encountered in ENGR 170 and haven’t used much since.  The format is LOOKUP Index, [Value0, Value1, …,ValueN], Variable.  The values inside the square brackets form an indexed list, with the index of the Value0 being 0, the index of Value1 being 1, and so forth.  For example, LOOKUP 2, [3, 2, 7, 8, 5], testval results in testval being set equal to 7.  The LOOKUP instruction is used in program FirstStepsForwardUsingTable as follows:

LOOKUP PrimitiveMoveIndex, [M1, M1, M2, M3, M4, M5, M6, M7, M8], PrimitiveMoveEEPROM  

We saw in part 3 above that M1 identifies EEPROM location 12, M2 location 20, and so forth.  Continuing the example from part 4 above, the first time LOOKUP is invoked PrimitiveMoveIndex is equal to 1.  So another way to interpret the LOOKUP instruction is:

LOOKUP 1, [12, 12, 20, 28, …..], PrimitiveMoveEEPROM

This will result in PrimitiveMoveEEPROM being set equal to 12, the EEPROM location where the data for movement M1 are stored.

The “Data Table” approach to walking is less easily understood than the “Brute Force” approach.  However it results in more compact code and a more efficient use of RAM and ROM in the microcontroller.  Make sure you understand thoroughly how FirstStepsForwardUsingTable works before continuing.
2. Modify FirstStepsForwardUsingTable to create a program called ThereAndBackUsingTable in which the robot walks forward 12 inches, then backs up the same distance.  Demonstrate operation, print a copy of your program, and add a schematic of your circuit.
Activity #4.  (5 pt.  Instructor’s initials ________) Walking forward – “State Transition” approach
In both the “Brute force” and the “Data Table” approaches to walking we have made use of 4 pieces of information:  The beginning and end points of the tilt servo rotation, and the beginning and end points of the stride servo rotation.  Each combination of these 4 variables yields a different primitive movement subroutine (in the case of the “brute force” method) or a different set of table entries (in the case of the “Data Table” method), which leads to a profusion of subroutines and/or table entries to cover all the possibilities.
There is a way to simplify.  Suppose you want to put the robot’s right foot forward.  With the “Brute force” and “Data table” approaches, there are 6 ways this can happen: With the robot tilted left, the stride servo can go from 1) center position to right forward position or 2) right back position to right forward position; With the robot tilted right, the stride servo can go from 3) center position to right forward position or 4) right back position to right forward position; With the robot not tilted, the stride servo can go from 5) center position to right forward position or 6) right back position to right forward position.  6 ways to do this means 6 subroutines or 6 table entries.  But do we really care what the starting position of the right foot is?  Probably not.  What we care about is where it finishes.  Also, each of the primitive moves rotates only 1 servo: either the tilt servo or the stride servo.  This means it may be possible to redefine the primitive movements to depend upon only 1 variable: the final position of the servo that is moving.  

In the “State Transition” approach we move only one servo at a time, we don’t make our moves from point A to point B.  We will simply move to point B.  Somehow we must be aware of what point A is, though, and act intelligently.  To do this we will introduce 2 new variables: CurrentTilt, and CurrentStride.

1. Load and run the following program. 

‘****************************************

'Program FirstStepsForwardUsingStateTransitions

' {$STAMP BS2}

' {$PBASIC 2.5}

'-------I/O Definitions ---------------

StrideServo CON       12

TiltServo   CON       13

'--------- Constant Declarations ------------

TiltStep  CON       5

StrideStep CON      5

MoveDelay CON       25

RightTilt CON       620

CenterTilt CON      750

LeftTilt CON        880

RightStride CON 650

CenterStride CON 750

LeftStride  CON 850

' The following are indices to be stored in EEPROM and used with the Branch instruction

'TL indicates "Tilt Left", SC indicates "Stride center" and so forth

TL  CON 0

TC  CON 1

TR  CON 2

SL  CON 3

SC  CON 4

SR  CON 5

xx  CON 255   'indicates end of steps in a basic move

'-----------Variable Declarations ----------

MoveLoop VAR Nib    'loop variable for normal walking

Pulses VAR Word

BasicMoveEEPROM VAR Word   'EEPROM location pointer for basic movement descriptions

PrimitiveMoveIndex VAR Byte    'Movement number (e.g. 1 = tilt from center to right, keep feet underneath)

CurrentTilt VAR Word            'Current position of tilt servo

NewTilt VAR Word              'Desired position of tilt servo

CurrentStride VAR Word         'Current position of stride servo

NewStride VAR Word             'Desired position of stride servo

'-----------EEPROM Data ------------------

'The following basic movements are composed of the listed primitive movements

WalkForward DATA TR, SL, TL, SR, xx  'primitive moves for normal walking

WalkBackward DATA TR, SR, TL, SL, xx

FinishWalk DATA TR, SC, TC, xx    'primitive movements for ending

'----- Main --------------

Main:

  GOSUB InitialCenter        'center both servos

  FOR MoveLoop = 1 TO 3  'take one full step each loop

    BasicMoveEEPROM = WalkForward

    GOSUB MakeBasicMove

  NEXT

  BasicMoveEEPROM = FinishWalk    'come to a stop with feet underneath

  GOSUB MakeBasicMove

END

'------Subroutines

InitialCenter:     'initial act - center both servos

  FOR Pulses = 1 TO 100 STEP StrideStep

    PULSOUT TiltServo, CenterTilt

    PULSOUT StrideServo, CenterStride

    PAUSE MoveDelay

  NEXT

  CurrentTilt = CenterTilt    'always end movement subroutine by saying where you are currently

  CurrentStride = CenterStride

  RETURN

MakeBasicMove:             'Step through movements listed in BasicMoveEEPROM

  READ BasicMoveEEPROM, PrimitiveMoveIndex

    BasicMoveEEPROM = BasicMoveEEPROM + 1    'increment the pointer to the EEPROM location for the next primitive

  IF PrimitiveMoveIndex = xx THEN RETURN  'end of the data, therefore return to main

  BRANCH PrimitiveMoveIndex, [TiltLeft, TiltCenter, TiltRight, StrideLeft, StrideCenter, StrideRight]

  RETURN 'return if none of the above apply (this should never happen)

  TiltLeft:

    NewTilt = LeftTilt

    NewStride = CurrentStride

    GOTO MakePrimitiveMove

 TiltCenter:

    NewTilt = CenterTilt

    NewStride = CurrentStride

    GOTO MakePrimitiveMove

  TiltRight:

    NewTilt = RightTilt

    NewStride = CurrentStride

    GOTO MakePrimitiveMove

  StrideLeft:

    NewStride = LeftStride

    NewTilt = CurrentTilt

    GOTO MakePrimitiveMove

  StrideCenter:

    NewStride = CenterStride

    NewTilt = CurrentTilt

    GOTO MakePrimitiveMove

  StrideRight:

    NewStride = RightStride

    NewTilt = CurrentTilt

    GOTO MakePrimitiveMove

  MakePrimitiveMove:

  IF CurrentTilt = NewTilt THEN

    FOR Pulses = CurrentStride TO NewStride STEP StrideStep    'hold tilt steady, do a stride

      PULSOUT TiltServo, CurrentTilt

      PULSOUT StrideServo, Pulses

      PAUSE MoveDelay

    NEXT

    CurrentStride = NewStride

  ELSE

    FOR Pulses = CurrentTilt TO NewTilt STEP TiltStep        'hold stride steady, do a tilt

      PULSOUT TiltServo, Pulses

      PULSOUT StrideServo, CurrentStride

      PAUSE MoveDelay

    NEXT

    CurrentTilt = NewTilt

  ENDIF

  GOTO MakeBasicMove         'loop until PrimitiveMoveIndex = xx

‘****************************************************************
Things to notice:

1. The new instruction here is the BRANCH instruction.  The syntax is BRANCH, Index, [Label0, Label1, …LabelN], where [Label0 …] is a list of labeled locations in the program, and Index is the index of the Labeled location the program should jump to.  For example, BRANCH, 2, [DoThis, DoThat, DoTheOther] will cause the program to jump to wherever DoTheOther is located in the program.  Note that these labels should not identify new subroutines because, unlike GOSUB, the BRANCH instruction does not expect control to return to the following line.
2. The 4 variables that define movement are CurrentTilt, CurrentStride, NewTilt, and NewStride.  The tilt servo rotates from CurrentTilt to NewTilt, and the stride servo rotates from CurrentStride to NewStride.  At the end of each primitive movement, CurrentTilt and CurrentStride are updated.
Understanding the “State Transition” approach should be fairly straightforward if you understood the “Data Table” approach.  Make sure that you do before continuing.  The rest of our work with the walking robot will use the “State Transition” approach.
2. Modify FirstStepsForwardUsingStateTransitions to create a program called JigDanceUsingStateTransitions in which the robot walks forward 12 inches, comes to a stop, does a little jig (rocks from one foot to the other) then backs up to the starting point.  Have the robot walk as quickly as possible without falling down.  Demonstrate operation, print a copy of your program, and add a schematic of your circuit.

